A procedure for finding the k-th power of a matrix

Branko Malesevic
Faculty of Electrical Engineering
University of Belgrade, Serbia
malesh@EUnet.yu

Ivana Jovovic
Faculty of Electrical Engineering
University of Belgrade, Serbia
ivana121@EUnet.yu

Introduction

This worksheet demonstrates the use of Maple in Linear Algebra.
We give a new procedure (**PowerMatrix**) in Maple for finding the k-th power of n-by-n square matrix A, in a symbolic form, for any positive integer k, k>=n. The algorithm is based on an application of Cayley-Hamilton theorem. We used the fact that the entries of the matrix A^k satisfy the same recurrence relation which is determined by the characteristic polynomial of the matrix A (see [1]). The order of these recurrences is n-d, where d is the lowest degree of the characteristic polynomial of the matrix A.
For non-singular matrices the procedure can be extended for k not only a positive integer.

Initialization

```
> restart:
  with(LinearAlgebra):
```

Procedure Definition

PowerMatrix

Input data are a square matrix **A** and a parameter **k**. Elements of the matrix **A** can be numbers and/or parameters. The parameter **k** can take numeric value or be a symbol. The output data is the k-th power of the matrix. The procedure PowerMatrix is as powerful as the procedure **rsolve**.

```
> PowerMatrix:=proc(A::Matrix,k)
  local i,j,m,r,q,n,d,f,P,F,C;
  P:=x->CharacteristicPolynomial(A,x);
  n:=degree(P(x),x);
```

```
d:=ldegree(P(x),x);
F:=(i,j)->rsolve({sum(coeff(P(x),x,m)*f(m+q),m=0..n)=0,seq
(f(r)=(A^r)[i,j],r=d+1..n)},f); C:=q->Matrix(n,n,F);
if type(k,integer) then return(simplify(A^k)) elif
(Determinant(A)=0 and not type(k,numeric)) then printf
("The %a-th power of the matrix for %a>=%d:",k,k,n) elif
(Determinant(A)=0 and type(k,numeric)) then return
(simplify(A^k)) fi; return(simplify(subs(q=k,C(q))));
end:
```

Examples

Example 1.

```
> A:=Matrix([[4,-2,2],[-5,7,-5],[-6,6,-4]]);
```

$$A := \begin{bmatrix} 4 & -2 & 2 \\ -5 & 7 & -5 \\ -6 & 6 & -4 \end{bmatrix} \quad (3.1.1)$$

```
> PowerMatrix(A,k);
```

$$\begin{bmatrix} 2\,3^k - 2^k & -2\,3^k + 2^{1+k} & 2\,3^k - 2^{1+k} \\ 5\,2^k - 5\,3^k & -4\,2^k + 5\,3^k & 5\,2^k - 5\,3^k \\ -6\,3^k + 6\,2^k & 6\,3^k - 6\,2^k & -6\,3^k + 7\,2^k \end{bmatrix} \quad (3.1.2)$$

```
> Determinant(A);
```

$$12 \quad (3.1.3)$$

```
> B:=A^(-1);
```

$$B := \begin{bmatrix} \frac{1}{6} & \frac{1}{3} & -\frac{1}{3} \\ \frac{5}{6} & -\frac{1}{3} & \frac{5}{6} \\ 1 & -1 & \frac{3}{2} \end{bmatrix} \quad (3.1.4)$$

```
> PowerMatrix(B,k);
```

$$\begin{bmatrix} -2^{-k} + 2\,3^{-k} & -2\,3^{-k} + 2^{1-k} & 2\,3^{-k} - 2^{1-k} \\ 5\,2^{-k} - 5\,3^{-k} & 5\,3^{-k} - 4\,2^{-k} & 5\,2^{-k} - 5\,3^{-k} \\ 6\,2^{-k} - 6\,3^{-k} & 6\,3^{-k} - 6\,2^{-k} & -6\,3^{-k} + 7\,2^{-k} \end{bmatrix} \quad (3.1.5)$$

Example 2.

```
> A:=Matrix([[1-p,p],[p,1-p]]);
```

$$A := \begin{bmatrix} 1-p & p \\ p & 1-p \end{bmatrix} \tag{3.2.1}$$

```
> PowerMatrix(A,k);
```

$$\begin{bmatrix} \frac{1}{2} + \frac{1}{2}\,(1-2\,p)^k & -\frac{1}{2}\,(1-2\,p)^k + \frac{1}{2} \\ -\frac{1}{2}\,(1-2\,p)^k + \frac{1}{2} & \frac{1}{2} + \frac{1}{2}\,(1-2\,p)^k \end{bmatrix} \tag{3.2.2}$$

The example is from [4], page 272, exercise 19.

Example 3.

```
> A:=Matrix([[a,b,c],[d,e,f],[g,h,i]]);
```

$$A := \begin{bmatrix} a & b & c \\ d & e & f \\ g & h & i \end{bmatrix} \tag{3.3.1}$$

```
> PowerMatrix(A,k)[1,1];
```

$$\sum_{\_R = RootOf\left(-1 + (g\,b\,f + h\,d\,c - g\,c\,e - h\,f\,a - i\,d\,b + i\,e\,a)\,\_Z^3 + (g\,c + h\,f - i\,e - i\,a + d\,b - e\,a)\,\_Z^2 + (i + e + a)\,\_Z\right)} \left(\left(1 - \_R\,i - \_R\,e - \_R^2\,h\,f + \_R^2\,i\,e\right)\left(\frac{1}{\_R}\right)^k\right) \Big/ \Big(\big(3\,\_R^2\,h\,d\,c - 3\,\_R^2\,e\,g\,c + 3\,\_R^2\,g\,b\,f + 3\,\_R^2\,i\,e\,a - 3\,\_R^2\,h\,f\,a + i + e + a + 2\,\_R\,g\,c + 2\,\_R\,h\,f - 2\,\_R\,i\,e - 2\,\_R\,i\,a + 2\,\_R\,d\,b - 2\,\_R\,e\,a - 3\,\_R^2\,i\,d\,b\big)\,\_R\Big) \tag{3.3.2}$$

Warning! In this example **MatrixPower** and **MatrixFunction** procedures cannot be done in real-time.

```
> # MatrixPower(A,k)[1,1];
> # MatrixFunction(A,v^k,v)[1,1];
```

Example 4.

```
> A:=Matrix([[0,0,1,0,1],[1,0,0,0,1],[0,0,0,1,1],[0,1,0,0,1],
  [1,1,1,1,0]]);
```

$$A := \begin{bmatrix} 0 & 0 & 1 & 0 & 1 \\ 1 & 0 & 0 & 0 & 1 \\ 0 & 0 & 0 & 1 & 1 \\ 0 & 1 & 0 & 0 & 1 \\ 1 & 1 & 1 & 1 & 0 \end{bmatrix} \quad (3.4.1)$$

```
> PowerMatrix(A,k)[1,5];
```

$$\frac{1}{17}\sqrt{17}\left(\left(\frac{1}{2}+\frac{1}{2}\sqrt{17}\right)^k-\left(\frac{1}{2}-\frac{1}{2}\sqrt{17}\right)^k\right) \quad (3.4.2)$$

Replace ':' with ';' and see result!

```
> MatrixPower(A,m)[1,5]:
> simplify(MatrixPower(A,m)[1,5]):
> assume(m::integer): simplify(MatrixPower(A,m)[1,5]):
```

The example is from [3], page 101.

Example 5. and Example 6.

Pay attention what happens for singular matrices.

Example 5.

```
> A:=Matrix([[0,2,1,3],[0,0,-2,4],[0,0,0,5],[0,0,0,0]]);
```

$$A := \begin{bmatrix} 0 & 2 & 1 & 3 \\ 0 & 0 & -2 & 4 \\ 0 & 0 & 0 & 5 \\ 0 & 0 & 0 & 0 \end{bmatrix} \quad (3.5.1.1)$$

```
> PowerMatrix(A,2);
```

$$\begin{bmatrix} 0 & 0 & -4 & 13 \\ 0 & 0 & 0 & -10 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \end{bmatrix} \quad (3.5.1.2)$$

```
> PowerMatrix(A,3);
```

(3.5.1.3)

$$\begin{bmatrix} 0 & 0 & 0 & -20 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \end{bmatrix} \tag{3.5.1.3}$$

```
> PowerMatrix(A,k);
```

The k-th power of the matrix for k>=4:

$$\begin{bmatrix} 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \end{bmatrix} \tag{3.5.1.4}$$

```
> MatrixPower(A,k);
```

$$LinearAlgebra:\text{-}LA\_Main:\text{-}MatrixPower\left(\begin{bmatrix} 0 & 2 & 1 & 3 \\ 0 & 0 & -2 & 4 \\ 0 & 0 & 0 & 5 \\ 0 & 0 & 0 & 0 \end{bmatrix}, k, outputoptions = [\,]\right) \tag{3.5.1.5}$$

```
> MatrixFunction(A,v^k,v);
```

Error, (in LinearAlgebra:-LA_Main:-MatrixFunction) could not compute finite interpolating value by evaluation of v^k*k/v at eigenvalue 0 which has multiplicity greater than one in the minimal polynomial

The example is from [2], page 151, exercise 23.

Example 6.

```
> A:=Matrix([[1,1,1,0],[1,1,1,-1],[0,0,-1,1],[0,0,1,-1]]);
```

$$A := \begin{bmatrix} 1 & 1 & 1 & 0 \\ 1 & 1 & 1 & -1 \\ 0 & 0 & -1 & 1 \\ 0 & 0 & 1 & -1 \end{bmatrix} \tag{3.5.2.1}$$

```
> PowerMatrix(A,k);
```

The k-th power of the matrix for k>=4:

$$\begin{bmatrix} 2^{-1+k} & 2^{-1+k} & \frac{1}{16}\,2^k\left((-1)^{1+k}+5\right) & \frac{1}{16}\,2^k\left((-1)^k-1\right) \\ 2^{-1+k} & 2^{-1+k} & \frac{5}{16}\,2^k\left((-1)^{1+k}+1\right) & \frac{1}{16}\,2^k\left(-1+5\,(-1)^k\right) \\ 0 & 0 & (-1)^k\,2^{-1+k} & (-1)^{1+k}\,2^{-1+k} \\ 0 & 0 & (-1)^{1+k}\,2^{-1+k} & (-1)^k\,2^{-1+k} \end{bmatrix} \qquad \textbf{(3.5.2.2)}$$

```
> MatrixPower(A,k);
```

$$LinearAlgebra{:-}LA\_Main{:-}MatrixPower\left(\begin{bmatrix} 1 & 1 & 1 & 0 \\ 1 & 1 & 1 & -1 \\ 0 & 0 & -1 & 1 \\ 0 & 0 & 1 & -1 \end{bmatrix}, k, outputoptions = [\,]\right) \qquad \textbf{(3.5.2.3)}$$

```
> MatrixFunction(A,v^k,v);
Error, (in LinearAlgebra:-LA_Main:-MatrixFunction) could not compute
finite interpolating value by evaluation of v^k*k/v at eigenvalue 0
which has multiplicity greater than one in the minimal polynomial
```

References

[1] Branko Malesevic. *Some combinatorial aspects of the composition of a set of functions.* NSJOM., 2006 (36), 3-9.
http://www.im.ns.ac.yu/NSJOM/Papers/36_1/NSJOM_36_1_003_009.pdf or http://arxiv.org/abs/math.CO/0409287 .

[2] John B. Johnston, G. Baley Price, Fred S. Van Vleck. *Linear Equations and Matrices.* Addison-Wesley, 1966.

[3] Carl D. Meyer. *Matrix Analysis and Applied Linear Algebra* . SIAM, 2001.

[4] Robert Messer. *Linear Algebra Gateway to Mathematics.* New York: Harper-Collins College Publisher, 1993.

Conclusions

This procedure has an educational character. It is an interesting demonstration for finding the k-th power of a matrix in a symbolic form. Sometimes, it gives solutions in the better form than the existing procedure **MatrixPower** (see example 4.). See also example 5. and example 6., where we consider singular matrices. In these cases the procedure **MatrixPower** does not give a solution. The procedure **PowerMatrix** calculates the k-th power of any singular matrix. In some examples it is possible to get a solution in the better form with using the procedure **allvalues** (see example 3.).

Acknowledgment: Research partially supported by MNTRS, Serbia, Grant No. 144020.